Example source using GMS81032

NEC format (uDP6121G)

INTRODUCTION

The GMS81004/08/16/24/32 are an advanced CMOS 8-bit microcontroller with 4K/8K/16K/24K/32K bytes of ROM which provides a highly flexible and cost effective solution to many universal remote control transmitter such as TV, VCR, CATV, DVD etc. The GMS81004/08/16/24/32 offers 448 bytes of on-chip data memory, 8-bit basic interval timer, 8-bit timer/counter, 16-bit timer/counter, 6-bit watchdog timer, power on reset, low voltage detection circuit. In addition, it supports power saving mode (stop mode) to reduce power consumption. Package types are provided with 20, 24, 28 SKDIP/SOP. The OTP (one time programmable ROM) device has the program memory locking function to prevent illegal copy and devices name are provided as a HMS81020TL/HMS81032TL

FEATURES

- 4K/8K/16K/24K/32K bytes on-chip program memory
- 448 bytes of on-chip data memory
- Minimum instruction execution time: 1uS at 4MHz (2cycle NOP Instruction)
- One 8-bit basic interval timer
- Two 8-bit Timer/Counter
- One 16-bit Timer/Counter
- Watchdog timer auto start (During 1second after power on reset)
- Power on reset
- Power saving mode (stop mode)
- Low voltage detection circuit
- Eight interrupt sources
  - External input: 2
  - Key scan:1
  - Timer: 5
- Operating voltage range
  - Mask : 2.2~4.0 V at 4MHz
  - OTP : 2.0~4.0 V at 4MHz

KEY NO.	DATA(H)	KEY NO.	DATA(H)
K01	00	K29	1C
K02	01	K30	1D

Table 1: Truth table for example program

KEY NO.	DATA(H)	KEY NO.	DATA(H)
K03	02	K31	1E
K04	03	K32	1F
K05	04	K33	20
K06	05	K34	21
K07	06	K35	22
K08	07	K36	23
K09	08	K37	24
K10	09	K38	25
K11	0A	K39	26
K12	0B	K40	27
K13	0C	K41	28
K14	0D	K42	29
K15	0E	K43	2A
K16	0F	K44	2B
K17	10	K45	2C
K18	11	K46	2D
K19	12	K47	2E
K20	13	K48	2F
K21	14	K49	30
K22	15	K50	31
K23	16	K51	32
K24	17	K52	33
K25	18	K53	34
K26	19	K54	35
K27	1A	K55	36
K28	1B	K56	37

Table 1: Truth table for example program

OUTPUT WAVEFORM FOR uPD6121G



Figure 1 OUTPUT WAVEFORM FOR uPD6121G

CIRCUIT DIAGRAM OF GMS81032



1. Tr. and IR LED should be located as far as possible from the IC.
2. Power capacitor and noise filtering capacitor should be located as near as possible from the IC.
3. Ceramic resonator should be located as near as possible from the IC.



Appendix A listing: example program code

Author: *Heejin Ryu*
MCU application design team
e-mail: heejin.ryu@hynix.com

APPENDIX A

```
1               ;*******************************************************************************
2               ;GMS81032 EXAMPLE PROGRAM WITH uPD6121G FORMAT
3               ;2002.6.19.           HeeJin Ryu
4
5               ;;;;;;;;;;;;;;;    REGISTER DEFINE ;;;;;;;;;;
6
7               R0       EQU     0C0H         ;PORT R0 REGISTER
8               R00      EQU     0,0C0H
9               R01      EQU     1,0C0H
10              R02      EQU     2,0C0H
11              R03      EQU     3,0C0H
12              R04      EQU     4,0C0H
13              R05      EQU     5,0C0H
14              R06      EQU     6,0C0H
15              R07      EQU     7,0C0H
16              R0DD     EQU     0C1H         ;PORT R0 DATA I/O DIRECTION REGISTER
17              R1       EQU     0C2H         ;PORT R1 REGISTER
18              R10      EQU     0,0C2H
19              R11      EQU     1,0C2H
20              R12      EQU     2,0C2H
21              R13      EQU     3,0C2H
22              R14      EQU     4,0C2H
23              R15      EQU     5,0C2H
24              R16      EQU     6,0C2H
25              R17      EQU     7,0C2H
26              R1DD     EQU     0C3H         ;PORT R1 DATA I/O DIRECTION REGISTER
27              R2       EQU     0C4H         ;PORT R2 REGISTER
28              R20      EQU     0,0C4H
29              R21      EQU     1,0C4H
30              R22      EQU     2,0C4H
31              R23      EQU     3,0C4H
32              R24      EQU     4,0C4H
33              R2DD     EQU     0C5H         ;PORT R2 DATA I/O DIRECTION REGISTER
34              CKCTLR   EQU     0C7H         ;CLOCK CONTROL REGISTER
35              BITR     EQU     0C7H         ;BASIC INTERVAL TIMER REGISTER
36              WDTR     EQU     0C8H         ;WATCHDOG TIMER REGISTER
37              PMR1     EQU     0C9H         ;PORT R1 MODE REGISTER
38              IMOD     EQU     0CAH         ;INTERRUPT MODE REGISTER
39              IEDS     EQU     0CBH         ;EXTERNAL INTERRUPT EDGE SELECTION
40              WDTCL    EQU     6,0C8H
41
42              IENL     EQU     0CCH         ;INTERRUPT ENABLE REGISTER LOW
43              IEWDT    EQU     6,0CCH       ;WATCHDOG TIMER INTERRUPT ENABLE FLAG
44              IEBIT    EQU     5,0CCH       ;BASIC INTERVAL TIMER INTERRUPT ENABLE FLAG
45
46              IRQL     EQU     0CDH         ;INTERRUPT REQUEST FLAG REGISTER LOW
47              IRQWDT   EQU     6,0CDH       ;WATCHDOG TIMER INTERRUPT REQUEST FLAG
48              IRQBIT   EQU     5,0CDH       ;BASIC INTERVAL TIMER INTERRUPT REQUEST FLAG
49
50              IENH     EQU     0CEH
51              IEKSCN   EQU     7,0CEH       ;EXTERNAL INT.0 ENABLE
52              IEE1     EQU     6,0CEH       ;EXTERNAL INT.1 ENABLE
53              IEE2     EQU     5,0CEH       ;EXTERNAL INT.2 ENABLE
54              IET0     EQU     3,0CEH       ;TIMER0 INT. ENABLE
55              IET1     EQU     2,0CEH       ;TIMER1 INT. ENABLE
56              IET2     EQU     1,0CEH       ;TIMER2 INT. ENABLE
57
58              IRQH     EQU     0CFH
59              IRQKSCN  EQU     7,0CFH       ;EXTERNAL INT.0 REQUST FLAG
60              IRQE1    EQU     6,0CFH       ;EXTERNAL INT.1 REQUST FLAG
61              IRQE2    EQU     5,0CFH       ;EXTERNAL INT.2 REQUST FLAG
```

```
62              IRQT0    EQU     3,0CFH       ;TIMER0 INT. REQUST FLAG
63              IRQT1    EQU     2,0CFH       ;TIMER1 INT. REQUST FLAG
64              IRQT2    EQU     1,0CFH       ;TIMER2 INT. REQUST FLAG
65
66              TM0      EQU     0D0H         ;TIMER0 MODE REGISTER
67              T0IFS    EQU     3,0D0H       ;TIMER0 INT EVE. 2ND SELECTION
68              T0MOD    EQU     4,0D0H       ;TIMER0 SING/MODULO-N
69              T0CN     EQU     5,0D0H       ;TIMER0 CONTINUOUS/PAUSE
70              T0ST     EQU     6,0D0H       ;TIMER0 START/STOP
71              CAP0     EQU     7,0D0H       ;TIMER0 TIMER.COUNTER/INPUT CAPTURE
72
73              TM1      EQU     0D1H
74              T1IFS    EQU     4,0D1H       ;TIMER1 INT EVE. 2ND SELECTION
75              T1MOD    EQU     5,0D1H       ;TIMER1 SING/MODULO-N
76              T1CN     EQU     6,0D1H       ;TIMER1 CONTINUOUS/PAUSE
77              T1ST     EQU     7,0D1H       ;TIMER1 START/STOP
78
79              TM2      EQU     0D2H
80              T2CN     EQU     3,0D2H       ;TIMER2 CONTINUOUS/PAUSE
81              T2ST     EQU     4,0D2H       ;TIMER2 SRATR/STOP
82
83              T0HMD    EQU     0D3H         ;TIMER0 HIGH-MSB DATA REGISTER
84              T0HLD    EQU     0D4H         ;TIMER0 HIGH-LSB DATA REGISTER
85              T0LMD    EQU     0D5H         ;TIMER0 LOW-MSB DATA REGISTER
86              T0LLD    EQU     0D6H         ;TIMER0 LOW-LSB DATA REGISTER
87
88              T1HD     EQU     0D7H         ;TIMER1 HIGH DATA REGISTER
89              T1LD     EQU     0D8H         ;TIMER1 LOW DATA REGISTER
90
91              T2DR     EQU     0D9H         ;TIMER2 DATA REGISTER
92              TM01     EQU     0DAH         ;TIMER0/TIMER1 MODE REGISTER
93              SMRR0    EQU     0DCH         ;STANDBY RELEASE REGISTER R0
94              SMRR1    EQU     0DDH         ;STANDBY RELEASE REGISEER R1
95              R1ODC    EQU     0DEH         ;PORT R1 OPEN DRAIN ASSIGN REGISTER
96
97
98              ;;;;;;;;; VECTOR TABLE ;;;;;;;;;;;;;;;;;;;;;;;
99
100                       ORG     0FFE6H
101 FFE6 92C1             DW      UNUSE        ;B.I.T
102 FFE8 92C1             DW      UNUSE        ;W.D.T
103 FFEA 92C1             DW      UNUSE        ;A/D CON
104 FFEC 92C1             DW      UNUSE        ;TIMER3
105 FFEE 92C1             DW      UNUSE        ;TIMER2
106 FFF0 92C1             DW      UNUSE        ;TIMER1
107 FFF2 92C1             DW      UNUSE        ;TIMER0
108 FFF4 92C1             DW      UNUSE        ;EXTERNAL INT 3
109 FFF6 92C1             DW      UNUSE        ;EXTERNAL INT 2
110 FFF8 92C1             DW      UNUSE        ;EXTERNAL INT 1
111 FFFA 92C1             DW      UNUSE        ;EXTERNAL INT 0
112 FFFC 92C1             DW      UNUSE        ;KEY SCAN
113 FFFE 00C0             DW      RESET
114
115
116              ;;;;;;;;;; MACRO DEFINE  ;;;;;;;;;;;;;;;;;;;;;;;;;;
117
118              SAVE     MACRO
119                       PUSH    A
120                       PUSH    X
121                       PUSH    Y
122                       PUSH    PSW
123                       ENDM
```

```
124
125                RESRORE  MACRO
126                         POP     PSW
127                         POP     Y
128                         POP     X
129                         POP     A
130                         ENDM
131
132                ;;;;;;;;;;;;;;RAM DEFINE;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
133
134                STAT     DS      1            ;STATUS RAM
135                NOKYF    EQU     0,STAT       ;NO KEY FLAG
136                SECONDF  EQU     1,STAT       ;KEY SCAN TWO TIMES CHECK FLAG
137                REPEATF  EQU     2,STAT       ;REPEAT CHECK FLAG
138
139                KEYNUM   DS      1            ;KEY NUMBER
140                NEWKEY   DS      1            ;PRESSED KEY POSITION
141                BACKEY   DS      1            ;BACK-UP KEY POSITION
142                OLDKEY   DS      1            ;ANOTHER BACK-UP KEY POSITION FOR REPEAT
143
144                TOTLKY   DS      1            ;NUMBER OF PRESSED KEY
145                PRTDAT   DS      1            ;TEMPORARY DATA
146                STROBE   DS      1            ;KEYSCAN STROBE
147                DLYCOUNT DS      1            ;DLYCOUNT RAM
148
149                DATA     DS      1            ;CSUTOM
150                DATA1    DS      1            ;CUSTOM BAR
151                DATA2    DS      1            ;DATA
152                DATA3    DS      1            ;DATA BAR
153
154                ;;;;;;;; MAIN PROGRAM  ;;;;;;;;;;;;;;;;;;;;;;
155
156                         ORG     0C000H       ;16K ROM START
157
158 C000 40        RESET:   CLRG
159 C001 1EFE              LDX     #0FEH        ;STACK POINTER INITIALIZE
160 C003 8E                TXSP                 ;SP. <--#0FEH
161 C004 3B3AC0            CALL    RAMCLEAR     ;RAM CLEAR
162
163 C007 3B75C1   NOKYR:   CALL    !REG_INI     ;REGISER INITIALIZE
164 C00A F1CF              CLR1    IRQKSCN      ;CLEAR KEYSCAN INT. REQUEST FLAG
165 C00C 00                STOP
166
167 C00D FF                NOP                  ;AFTER WAKE-UP TWO NOP INSTRUCTION ARE NEEDED
168 C00E FF                NOP
169
170 C00F E4FFC2   MAIN:    LDM     R1,#0FFH     ; OUTPUT HIGH
171 C012 E40003            LDM     BACKEY,#0
172 C015 E40004            LDM     OLDKEY,#0
173 C018 E40000            LDM     STAT,#0
174                MAIN1:
175 C01B 3B18C1            CALL    !LINESCAN    ;KEY SCAN
176 C01E E47FC8            LDM     WDTR,#7FH    ;WDT CLEAR & THE VALUE 3FH
177                ;INTERVAL OF WDT= 3FH*16.384uS=1032MS
178 C021 C432              LDA     #50
179 C023 3B6DC1   DELAYAG: CALL    !DELAY       ;TIME DELAY FOR 2ND KEY SCANNING
180 C026 A8                DEC     A
181 C027 70FA              BNE     DELAYAG
182
183 C029 C505     KEYCHK:  LDA     TOTLKY
184 C02B F006              BEQ     NOKYROUT     ;NO KEY
185 C02D 4401              CMP     #1
```

```
186 C02F F013             BEQ     NORMALR      ;ONE KEY
187 C031 2FDC             BRA     MAIN         ;DOUBLE KEY
188
189 C033 0300CA   NOKYROUT: BBS     NOKYF,RESET
190 C036 0100             SET1    NOKYF
191 C038 2FE1             BRA     MAIN1
192
193 C03A C400     RAMCLEAR: LDA     #0
194 C03C 1E00             LDX     #0
195 C03E FB       RAMCLR:  STA     {X}+         ;RAM CLEAR
196 C03F 5EC0             CMPX    #0C0H        ;0000H~00BFH
197 C041 70FB             BNE     RAMCLR
198 C043 6F               RET
199
200 C044 330006   NORMALR: BBC     SECONDF,SAVEKEY
201
202 C047 C502             LDA     NEWKEY       ;COMPARE 1ST KEY WITH 2ND KEY
203 C049 4503             CMP     BACKEY
204 C04B F009             BEQ     OKMOD1
205
206 C04D 2100     SAVEKEY: SET1    SECONDF
207 C04F C502             LDA     NEWKEY       ;SAVE 1ST KEY
208 C051 E503             STA     BACKEY
209 C053 1B1BC0           JMP     MAIN1
210
211 C056 C503     OKMOD1:  LDA     BACKEY       ;REPEAT CHECK
212 C058 4504             CMP     OLDKEY
213 C05A F002             BEQ     OKMOD2
214 C05C 5100             CLR1    REPEATF
215 C05E E504     OKMOD2:  STA     OLDKEY
216
217 C060 E40409           LDM     DATA,#04H    ;CUSTOM
218 C063 E4FB0A           LDM     DATA1,#0FBH  ;CUSTOM BAR
219 C066 C502             LDA     NEWKEY
220 C068 9F               TAY
221 C069 D594C1           LDA     !NEC_DATA+Y
222 C06C E50B             STA     DATA+2       ;DATA
223 C06E E50C             STA     DATA+3
224 C070 2C0C             COM     DATA+3       ;DATA BAR
225
226 C072 E424D7           LDM     T1HD,#24H    ; CARRIER DEFINE,250nS*24H=9000nS
227 C075 E446D8           LDM     T1LD,#46H    ; 250nS*46H=17500nS
228
229 C078 E423D3           LDM     T0HMD,#23H   ; HEADER DEFINE,1uS*2328H=9000uS
230 C07B E428D4           LDM     T0HLD,#28H
231 C07E E411D5           LDM     T0LMD,#11H   ; 1194H*1uS=4500uS
232 C081 E494D6           LDM     T0LLD,#94H
233
234
235 C084 E400DA           LDM     TM01,#00H
236 C087 E400D0           LDM     TM0,#00H
237 C08A E400D1           LDM     TM1,#00H
238
239 C08D 71CF             CLR1    IRQT0
240 C08F 51CF             CLR1    IRQT1
241
242 C091 530008           BBC     REPEATF,FULL_DATA
243 C094 33CFFD           BBC     IRQT2,$      ;108mS WAITING
244 C097 31CF             CLR1    IRQT2
245 C099 1BFCC0           JMP     RPT_DATA
246
247 C09C E400D2   FULL_DATA: LDM     TM2,#00H
```

```
248 C09F 31CF              CLR1    IRQT2
249 C0A1 E4D3D9             LDM     T2DR,#0D3H    ;TIMER2, 512uS*D3H=108.032uS
250 C0A4 E41ED2            LDM     TM2,#0001_1110B   ;TIMER2 START AFTER CLEAR,CONTI,PS11,512uS
251
252 C0A7 E4D0D1            LDM     TM1,#1101_0000B   ;TIMER START AFTER CLEAR,CONTI,MODULO,
253                                                  ;INT. EVERY 2ND COUNTER, PS0 ,250nS
254 C0AA E46AD0            LDM     TM0,#0110_1010B   ;TIMER,TIMER0 START AFTER CLEAR,CONTI,
255                                                  ;MODULO,2ND, PS2, 1uS
256 C0AD E4CCDA            LDM     TM01,#1100_1100B  ;TOUT LOGIC OUTPUT, REMOUT OUTPUT HIGH,
257                                                  ;TIMER0 OUTPUT HIGH, TIMER1 OUTPUT HIGH,
258                                                  ;AND OF TO & T1
259 C0B0 73CFFD             BBC     IRQT0,$       ; HEAD OUTPUT
260 C0B3 71CF              CLR1    IRQT0
261
262 C0B5 1E00              LDX     #0           ;BYTE COUNTER
263 C0B7 3E00              LDY     #0           ;BIT COUNTER
264 C0B9 E402D3            LDM     T0HMD,#02H   ;MARK PULSE 560uS
265 C0BC E430D4            LDM     T0HLD,#30H
266
267
268 C0BF 3BDFC0   PULSE_DATA: CALL    !PULSE_OUT
269 C0C2 3E00              LDY     #0
270 C0C4 8F                INC     X
271 C0C5 5E04              CMPX    #4
272 C0C7 70F6              BNE     PULSE_DATA
273
274 C0C9 E462D0   END_PULSE: LDM     TM0,#0110_0010B   ;2ND-->EVE
275 C0CC 73CFFD            BBC     IRQT0,$
276 C0CF 71CF              CLR1    IRQT0
277 C0D1 E400DA            LDM     TM01,#00H    ;REMOUT END
278 C0D4 E400D0            LDM     TM0,#00H
279 C0D7 E400D1            LDM     TM1,#00H
280 C0DA 4100              SET1    REPEATF
281 C0DC 1B1BC0            JMP     MAIN1
282
283 C0DF 7909     PULSE_OUT: ROR     DATA+X
284 C0E1 5008              BCC     LOW_BIT      ;BIT0
285
286 C0E3 E406D5   HIGB_BIT: LDM     T0LMD,#06H  ;1690uS
287 C0E6 E49AD6            LDM     T0LLD,#9AH
288 C0E9 2F06              BRA     COMEND
289
290 C0EB E402D5   LOW_BIT: LDM     T0LMD,#02H   ;565uS
291 C0EE E435D6            LDM     T0LLD,#35H
292
293 C0F1 73CFFD   COMEND:  BBC     IRQT0,$
294 C0F4 71CF              CLR1    IRQT0
295 C0F6 9E                INC     Y
296 C0F7 7E08              CMPY    #8
297 C0F9 70E4              BNE     PULSE_OUT
298 C0FB 6F                RET
299
300 C0FC E408D5   RPT_DATA: LDM     T0LMD,#08H       ;HEAD LOW CHANGE 4.5-->2.25
301 C0FF E4CAD6            LDM     T0LLD,#0CAH
302 C102 E4D0D1            LDM     TM1,#1101_0000B   ;TIMER START AFTER CLEAR,CONTI,MODULO,
303                                                  ;INT. EVERY 2ND COUNTER, PS0 ,250nS
304 C105 E46AD0            LDM     TM0,#0110_1010B   ;TIMER,TIMER0 START AFTER CLEAR,CONTI,
305                                                  ;MODULO,2ND, PS2, 1uS
306 C108 E4CCDA            LDM     TM01,#1100_1100B  ;TOUT LOGIC OUTPUT, REMOUT OUTPUT HIGH,
307                                                  ;TIMER0 OUTPUT HIGH, TIMER1 OUTPUT HIGH,
308                                                  ;AND OF TO & T1
309 C10B 73CFFD            BBC     IRQT0,$
```

```
310 C10E 71CF              CLR1    IRQT0
311 C110 E402D3            LDM     T0HMD,#02H   ;MARK PULSE 560uS
312 C113 E430D4            LDM     T0HLD,#30H
313 C116 2FB1              BRA     END_PULSE
314
315                ;;;;;;;;;;;;;;;;;;;;;;;;;;
316 C118 E40001    LINESCAN: LDM     KEYNUM,#0
317 C11B E40002            LDM     NEWKEY,#0
318 C11E E40006            LDM     PRTDAT,#0
319 C121 E40005            LDM     TOTLKY,#0
320 C124 E40007            LDM     STROBE,#0
321 C127 E40008            LDM     DLYCOUNT,#0
322
323 C12A E4FE07            LDM     STROBE,#1111_1110B
324 C12D C507      KEYAG:  LDA     STROBE
325 C12F E5C2              STA     R1
326 C131 3B6DC1            CALL    !DELAY
327 C134 C5C0              LDA     R0           ;READ INPUT
328 C136 E4FFC2            LDM     R1,#0FFH
329 C139 E506              STA     PRTDAT       ;R0 INPUT IMSI DATA
330 C13B 44FF              CMP     #0FFH
331 C13D F002              BEQ     NOKYPRE
332 C13F 2F18              BRA     SAVEDATA
333
334 C141 20        NOKYPRE: CLRC
335 C142 C408              LDA     #8
336 C144 0501              ADC     KEYNUM
337 C146 E501              STA     KEYNUM       ;KEYNUM=KEYNUM+8
338
339 C148 C4FF      NEXTAG: LDA     #0FFH
340 C14A E5C2              STA     R1
341 C14C 3B6DC1            CALL    !DELAY
342 C14F A0                SETC
343 C150 2907              ROL     STROBE
344 C152 C507              LDA     STROBE
345 C154 447F              CMP     #7FH
346 C156 70D5              BNE     KEYAG
347 C158 6F                RET
348
349 C159 1E08      SAVEDATA: LDX     #8
350 C15B A0        SAVEDT1: SETC
351 C15C 6906              ROR     PRTDAT
352 C15E D006              BCS     INCKEY       ; THIS IS NOT PRESSED
353
354 C160 8905              INC     TOTLKY       ; THIS IS PRESSED
355 C162 C501              LDA     KEYNUM
356 C164 E502              STA     NEWKEY
357
358 C166 8901      INCKEY: INC     KEYNUM
359 C168 AF                DEC     X
360 C169 70F0              BNE     SAVEDT1
361 C16B 2FDB              BRA     NEXTAG
362
363
364 C16D E40A08    DELAY:  LDM     DLYCOUNT,#10
365 C170 A908      CONDLY: DEC     DLYCOUNT
366 C172 70FC              BNE     CONDLY
367 C174 6F                RET
368
369 C175 60        REG_INI: DI
370 C176 E43DC7            LDM     CKCTLR,#0011_1101B   ;WDT-ON,ENPCK-ON,16uS STANDBY RELEASE TIME (PS8)
371 C179 E4FFDE            LDM     R1ODC,#1111_1111B   ;R1 PORT OPEN DRAIN.
```

```
372 C17C E4FFC3            LDM     R1DD,#1111_1111B   ;R1 OUTPUT
373 C17F E400C1            LDM     R0DD,#0000_0000B   ;R0 INPUT
374 C182 E480CE            LDM     IENH,#1000_0000B   ;KEY SCAN
375 C185 E400CC            LDM     IENL,#0000_0000B
376 C188 E400CF            LDM     IRQH,#0000_0000B
377 C18B E4FFDC            LDM     SMRR0,#1111_1111B   ;R0 STANDBY RELEASE
378 C18E E400C2            LDM     R1,#0000_0000
379 C191 6F                RET
380
381 C192 FF       UNUSE:   NOP
382 C193 7F                RETI
383
384                ;;;;;;NEC DATA LOOK UP TABLE
385                NEC_DATA:
386 C194 00                DB      00H,01H,02H,03H,04H,05H,06H,07H
    C195 01
    C196 02
    C197 03
    C198 04
    C199 05
    C19A 06
    C19B 07
387 C19C 08                DB      08H,09H,0AH,0BH,0CH,0DH,0EH,0FH
    C19D 09
    C19E 0A
    C19F 0B
    C1A0 0C
    C1A1 0D
    C1A2 0E
    C1A3 0F
388 C1A4 10                DB      10H,11H,12H,13H,14H,15H,16H,17H
    C1A5 11
    C1A6 12
    C1A7 13
    C1A8 14
    C1A9 15
    C1AA 16
    C1AB 17
389 C1AC 18                DB      18H,19H,1AH,1BH,1CH,1DH,1EH,1FH
    C1AD 19
    C1AE 1A
    C1AF 1B
    C1B0 1C
    C1B1 1D
    C1B2 1E
    C1B3 1F
390 C1B4 20                DB      20H,21H,22H,23H,24H,25H,26H,27H
    C1B5 21
    C1B6 22
    C1B7 23
    C1B8 24
    C1B9 25
    C1BA 26
    C1BB 27
391 C1BC 28                DB      28H,29H,2AH,2BH,2CH,2DH,2EH,2FH
    C1BD 29
    C1BE 2A
    C1BF 2B
    C1C0 2C
    C1C1 2D
    C1C2 2E
    C1C3 2F
```

```
392 C1C4 30                DB      30H,31H,32H,33H,34H,35H,36H,37H
    C1C5 31
    C1C6 32
    C1C7 33
    C1C8 34
    C1C9 35
    C1CA 36
    C1CB 37
393 C1CC 38                DB      38H,39H,3AH,3BH,3CH,3DH,3EH,3FH
    C1CD 39
    C1CE 3A
    C1CF 3B
    C1D0 3C
    C1D1 3D
    C1D2 3E
    C1D3 3F
394
395                        END                 ;NOTICE PROGRAM END


        -- 0 Error(s) --

   --- Total Machine Code : 494 Bytes --
```